

Description

`frame putlabel` copies a list of value labels from the current frame to other frames.

Quick start

Copy value labels `v11`, `v12`, and `v13` from the current frame into frame `fr1`

```
frame putlabel v11 v12 v13, into(fr1)
```

Copy all value labels from the current frame into frames `fr2` and `fr3`

```
frame putlabel _all, into(fr2 fr3)
```

Menu

Data > Frames Manager

Syntax

```
frame putlabel lblnames, into(framenames) [replace]
```

Options

`into(framenames)` specifies that the value labels be copied to the frames *framenames*. `into()` is required.

`replace` allows existing value labels in the destination frames to be replaced by value labels with the same name from the current frame.

Remarks and examples

When you use `frame put` to copy variables into a new frame, the value labels attached to the variables are also copied to the new frame. However, sometimes you need to copy only the value labels from one frame to another, and there are two methods for doing so. `label copy` allows you to copy a value label from one frame to another frame. `frame putlabel` allows you to copy multiple value labels from the current frame to one or more other frames.

► Example 1

Suppose our Stata session has multiple frames that are populated with data, but only the current frame has value-label definitions.

Here are the value labels in the current frame.

```
. label list
leaning:
      1 Liberal
      2 Moderate
      3 Conservative

agecat:
      0 1-5
      2 6-12
      3 13-19
      4 20-29
      5 30-39
      6 40-49
      7 50+

logic:
      0 False
      1 True

sex:
      0 Male
      1 Female

onoff:
      0 Off
      1 On
```

Here are the frames in our session.

```
. frames dir
* default  50 x 5
* f1       100 x 7
* f2       200 x 8
* f3       300 x 4
* f4       400 x 6
* f5       410 x 13
```

Note: Frames marked with * contain unsaved data.

Let's copy the agecat value labels into frame f1.

```
. frame putlabel agecat, into(f1)
. frame f1: label list
agecat:
      0 1-5
      2 6-12
      3 13-19
      4 20-29
      5 30-39
      6 40-49
      7 50+
```

This is equivalent to

```
. label copy agecat agecat, toframe(f1)
```

The following uses the `_all` shortcut to copy all the value labels from the current frame to the other frames, then lists the names of the value labels defined in frame `f5`.

```
. frame putlabel _all, into(f2 f3 f4 f5)
. frame f5: label dir
onoff
sex
logic
agecat
leaning
```

It is up to you to attach the copied value labels to the variables in the target frames.



Also see

[D] [frames intro](#) — Introduction to frames

[D] [frames](#) — Data frames

[D] [label](#) — Manipulate labels

